
Abstract

Proposing relevant perturbations to biological signaling networks is central to many problems
in biology and medicine because it allows for enabling or disabling certain biological outcomes.
In contrast to quantitative methods that permit fine-grained (kinetic) analysis, qualitative ap-
proaches allow for addressing large-scale networks. This is accomplished by more abstract rep-
resentations such as logical networks. We elaborate upon such a qualitative approach aiming at
the computation of minimal interventions in logical signaling networks relying on Kleene’s three-
valued logic and fixpoint semantics. We address this problem within answer set programming
and show that it greatly outperforms previous work using dedicated algorithms.

1 Introduction

Systems biology is a field at the crossover of biology, informatics, and mathematics. It

aims at developing methods and models to elucidate the functioning of biological systems.

Among them, signaling networks are crucial for the understanding of the fast response of a

system to external perturbations. Importantly, they are involved in bio-medical processes

and their control has a crucial impact on drug target identification and diagnosis.

During the last decade, many efforts have been made to develop relevant formalisms

and modeling frameworks to take into account the specificities of such biological systems.

In the lack of quantitative details, qualitative approaches, such as Boolean logical net-

works (Kauffman 1969; Thomas 1973), have become very popular (Wang et al. 2012). It

has been proved that the early response of signaling networks can be appropriately mod-

eled with Boolean logical networks, as illustrated on several signal transduction pathways

involved in diverse processes such as proliferation, cell cycle regulation, apoptosis, or dif-

ferentiation (Saez-Rodriguez et al. 2007; Samaga et al. 2009; Saez-Rodriguez et al. 2011;

Calzone et al. 2010). Moreover, inference of Boolean networks from experimental data

can now be performed with several computational methods (Saez-Rodriguez et al. 2009;

Mitsos et al. 2009; Sharan and Karp 2012; Videla et al. 2012).

As mentioned in the seminal paper for systems biology (Kitano 2002), a major chal-

lenge in this field is how to systematically control the state of the cell. From an application

viewpoint, this means selecting appropriate drugs in order to force the system to reach a

steady state with properties that were specified a priori. Thus, progress in this area would

lead to hypothesis-driven research in biology. Nowadays, due to the lack of information,

multiple hypotheses are usually generated from prior knowledge and computational mod-

els. Next, decision-making methods can be used to suggest new experiments in order to

reduce ambiguous hypotheses. Finally, new experimental data is produced to test the

generated hypotheses, the models are refined, and the loop is started over again (Kitano

2002; Kreutz and Timmer 2009; Sparkes et al. 2010).

The problem of identifying “key-players” in biological systems has been addressed

for metabolic, gene and signaling networks. However, the underlying mathematical for-

malisms for each of these biological networks allows for different computational ap-

proaches. Moreover, while significant progress has been made for metabolic (Kauffman

et al. 2003; Klamt 2006; Acuña et al. 2012; Stelling et al. 2002) and gene regulatory

networks (Bouaynaya et al. 2011; Faryabi et al. 2008; Karlebach and Shamir 2009),

controlling mechanisms in signaling networks remain poorly understood and only a few

approaches can be found in the literature (Abdi et al. 2008; Samaga et al. 2010; Wang

and Albert 2011). More precisely, apart from numerical methods (e.g. ordinary differ-

ential equations), computational formalisms for metabolic networks are based on linear

algebra whereas the dynamics in gene regulatory networks are well captured either by

probabilistic approaches or by discrete dynamical systems taking into account non-linear

effects (Batt et al. 2008; Naldi et al. 2009). On the contrary, the nature of signaling trans-

duction networks is closely related to that of digital circuits (?; ?; Abdi et al. 2008). Thus,

logic-based models are particularly well suited to describe and study the (early)-response

of such systems.

Among the few approaches addressing the challenge of controlling the state of the

cell in the context of signaling transduction, we focus in what follows on (Samaga et al.

2010). Based on earlier work (Klamt 2006) on metabolic networks, the notion of mini-

mal intervention sets was introduced in (Samaga et al. 2010) and dedicated algorithms

were developed to compute them. Intuitively, an “intervention” consists of an inclusion

minimal set of knock-ins (activation drugs) and knock-outs (inhibition drugs) that force

a set of target species or compounds into a desired state. Unfortunately, the dedicated

algorithms are computationally demanding due to the highly combinatorial mechanisms

in signaling networks. Therefore, they are limited to compute small intervention sets

and fail to scale over large-scale networks. In general, multiple interventions are neces-

sary to cope with robustness and cellular complexity (Stelling et al. 2004). Moreover,

previous work on the inference of logical networks suggests that, if the inherent noise

is considered, there are multiple networks compatible with the experimental observa-

tions (Saez-Rodriguez et al. 2009). Concretely, the mentioned limitations make it hard

to prove that the identified solutions are biologically robust to small perturbations of the

system or its environment. Thus, in order to overcome such limitations, more elaborate

and more powerful computational methods are needed towards large-scale systems and

robust solutions.

The question of scalability of computational methods for the identification or the

pruning of biological systems has already been successfully addressed with Answer Set

Programming (ASP; (Baral 2003)) in various settings, among them (Baral et al. 2004;

Erdem and Türe 2008; Gebser et al. 2011; Gebser et al. 2010; Ray et al. 2010; Videla et al.

2012). Of special interest is here ASP’s expressive power to address problems of elevated

complexity, in particular, for computing inclusion minimal models. In the same direction

as these papers, we provide a precise characterization of the minimal intervention set

problem relying on Kleene’s three-valued logic and fixpoint semantics (similar to that

of (Fitting 1985)). We introduce an ASP encoding to solve this problem and we evaluate

its performance on four real-world and biologically relevant benchmarks. 1 Interestingly,

our fixpoint based characterisation of biological network response is in line with the in-

vestigations of (Inoue 2011; Inoue and Sakama 2012) using different logic programming

semantics for characterising systems behaviour. To be more precise, Inoue uses in (Inoue

2011) a two-valued logic along with two-valued fixpoint semantics to characterise trajec-

tories and stable states of Boolean networks (by translating them into logic programs).

Hence such logic programming concepts appear to be appropriate tools for characteris-

ing these types of networks. Moreover, Boolean constraint solving technologies offer a

powerful computational framework to face the highly combinatorial nature of signaling

networks.

In what follows, we assume some familiarity with ASP, its semantics as well as its basic

language constructs. A comprehensive treatment of ASP can be found in (Baral 2003;

Gebser et al. 2012). Our encodings are written in the input language of gringo 3 (Gebser

et al. ).

2 Intervention set strategies

This section provides a formal characterization of intervention strategies, as treated

in (Samaga et al. 2010). In doing so, we follow the popular approach to qualitative mod-

eling in biology (Kauffman 1969; Thomas 1973) in representing biological networks as

logical networks2 and associating biological species or compounds (eg. receptors, kinases

or phosphatases) with propositional variables. More formally, a logical network consists

of a finite set V of propositional variables and a function φ mapping each variable v ∈ V
to a propositional formula φ(v) over V . We form propositional formulas from V with the

connectives ⊥, >, ¬, ∨, and ∧ in the standard way. For illustration, let us consider the

logical network reproduced from (Samaga et al. 2010) in Figure 1. This network consists

i1 i2

a

b

c

∨
d e

∨
f

g

o1 o2

∧∧

Fig. 1. Exemplary directed hypergraph representation of a logical network. We refer the reader
to (Klamt et al. 2009) for more details on hypergraphs and cellular networks.

of the set V of species variables {i1, i2, a, . . . , g, o1, o2} along with the function φ defined

1 The encodings are available at: http://potassco.sourceforge.net/apps.html#interventions
2 We refrain from using the term Boolean network in view of our usage of three-valued semantics.

as:

φ =

 i1 7→ i1
i2 7→ i2

a 7→ ¬d
b 7→ a ∧ i1
c 7→ b ∨ e

d 7→ c

e 7→ ¬i1 ∧ i2
f 7→ e ∨ g

g 7→ f

o1 7→ c

o2 7→ g


Note that φ leaves the specification of the (input) variables i1 and i2 open.

The steady states of such a network are given by truth assignments yielding identical

values for v and φ(v) for all v ∈ V (although not all are biological meaningful as we see be-

low). Following (Samaga et al. 2010), we adapt a three-valued setting relying on Kleene’s

three-valued logic (Kleene 1950) and thus consider truth assignments mapping formulas

to truth values {t,f ,u} according to Kleene’s semantics. Clearly, two-valued assignments

are restricted to range {t,f}. Observe that any logical network has a trivial three-valued

steady state obtained by assigned u to all variables. In fact, the major constituents of

intervention strategies can be captured by means of partial two-valued assignments (des-

ignating the absence or presence of certain species). We sometimes represent assignments

extensionally as sets, viz. {v 7→ A(v) | v ∈ V }, for checking containment, difference, etc.

To avoid conflicts when composing assignments, we define A ◦ B = (A \ B) ∪ B where

B = {v 7→ s | v 7→ s ∈ B} and t = f , f = t, u = u.

For capturing the dynamics of a logical network (V, φ), we define the following operator

on truth assignments over V : 3

Ω(V,φ)(A) = {v 7→ A(φ(v)) | v ∈ V }.

Among the (three-valued) steady states of (V, φ), we are interested in the fixpoint of

Ω(V,φ) reachable from the “undefined assignment”. As with Fitting’s operator (1985),

this fixpoint is unique and can be computed in polynomial time.

To this end, we define the iterative variant of Ω(V,φ) as

Ω0
(V,φ)(A) = A and Ωj+1

(V,φ)(A) = Ω(V,φ)

(
Ωj(V,φ)(A)

)
.

In biological terms, a sequence (Ωj(V,φ)(A))j∈J represents the evolution of a system start-

ing in state A.

For capturing modifications to a logical network, (Samaga et al. 2010) puts forward

the notion of clamping variables to Boolean values, thereby overriding their original

specification: The logical network (V, φ|A) is obtained from (V, φ) by clamping assignment

A over V , if

φ|A(v) =


> if A(v) = t

⊥ if A(v) = f

φ(v) otherwise

Given a network, the aim of an intervention strategy is to identify an intervention that

leads to a steady state satisfying a given goal under some side constraints. The concepts

of an intervention (I), goal (G), and side constraints (C) can be captured as partial

two-valued assignments (indicating the absence or presence of certain species).

To be more precise, given a logical network (V, φ), an intervention scenario is a pair

(G,C) of partial two-valued assignments over V and an intervention set is a partial

two-valued assignment I over a set of intervention variables X ⊆ V .

3 The interested reader may notice the resemblance to Fitting’s three-valued operator (Fitting 1985).

Intervention Strategy

Let (V, φ) be a logical network, let (G,C) be an intervention scenario, and X ⊆ V be

a set of intervention variables.

An intervention set I over X is an intervention strategy for (G,C) wrt (V, φ),

if for some j ≥ 0, we have

1. Ωj(V,φ|C◦I)
(Su) = Ωj+1

(V,φ|C◦I)
(Su) where Su = {v 7→ u | v ∈ V } and

2. G ⊆ Ωj(V,φ|C◦I)
(Su)

In words, Ωj(V,φ|C◦I)
(Su) is a steady state of the clamped network (V, φ|C◦I) satisfying

the goal conditions in G. The important biological property of Ωj(V,φ|C◦I)
(Su) is that each

of its variables remains undefined unless there is a cause to make it true or false.

The intervention set problem consists in deciding whether there is an intervention

strategy for an intervention scenario (G,C) wrt a logical network (V, φ). Roughly speak-

ing, the intervention set problem is a typical problem in NP : Once an intervention is

guessed, it can be verified in polynomial time.

As an example, consider the above logical network along with the intervention scenarios

(G1, C1) = ({o1 7→ f , o2 7→ t}, {i1 7→ t}) and (G2, C2) = ({a 7→ t}, ∅). The intervention

set {b 7→ f , e 7→ f , f 7→ t} satisfies both scenarios yielding the two steady states,

respectively:

{i1 7→ t, i2 7→ u, a 7→ t, b 7→ f , c 7→ f , d 7→ f , e 7→ f , f 7→ t, g 7→ t, o1 7→ f , o2 7→ t}
{i1 7→ u, i2 7→ u, a 7→ t, b 7→ f , c 7→ f , d 7→ f , e 7→ f , f 7→ t, g 7→ t, o1 7→ f , o2 7→ t}.

Now, let us define further intervention strategies relying on a finite family (Gj , Cj)j∈J
of intervention scenarios and k some positive integer.

• A multi-scenario intervention strategy for (Gj , Cj)j∈J wrt (V, φ) is an intervention

strategy for each (Gj , Cj) wrt (V, φ) for each j ∈ J .

• A bounded intervention strategy for (Gj , Cj)j∈J wrt (V, φ) and k is a multi-scenario

intervention strategy for (Gj , Cj)j∈J wrt (V, φ) of cardinality k′ ≤ k.

• A minimal bounded intervention strategy for (Gj , Cj)j∈J wrt (V, φ) and k is a

⊆-minimal multi-scenario intervention strategy for (Gj , Cj)j∈J wrt (V, φ) of cardi-

nality k′ ≤ k.

In the following, we are particularly interested in enumerating all minimal bounded

intervention strategies. 4 We apply two different approaches to ASP solving. The first

is claspD, which can enumerate all subset minimal solutions in polynomial space. The

second one is hclasp, which utilizes (heuristic-driven) solution recording and hence runs in

exponential space. Given that the intervention set problem has a potentially exponential

number of solutions, both algorithms run in exponential time. However, by using claspD,

it is possible to ask more complex queries. We can in principle put additional constraints

on the solution candidates enumerating only subset minimal solutions that have a certain

property. This is not possible with hclasp without embedding it into another algorithm

(and then not in polynomial space).

4 In (Samaga et al. 2010), minimal bounded intervention strategies are called minimal intervention sets.

3 Encoding

3.1 Instance representation

Let (V, φ) be a logical network. We represent the variables V as facts over predicate

variable/1, namely variable(v) for all v ∈ V . Facts over predicate candidate/1

denote the intervention variables that can be part of an intervention set. This allows

us to control on which species interventions are permitted, for example one can exclude

interventions over constrained or goal variables.

Without loss of generality, we assume that all formulas mapped by φ are in dis-

junctive normal form. Hence, φ(v) is a set of (dual) clauses and a clause a set of lit-

erals. We represent formulas using predicates formula/2, dnf/2, and clause/3. The

facts formula(v,sφ(v)) map variables v ∈ V to their corresponding formulas φ(v), facts

dnf(sφ(v),sc) associate φ(v) with its clauses c ∈ φ(v), facts clause(sc,v,1) associate

clause c with its positive literals v ∈ c∩V , and facts clause(sc,v,-1) associate clause c

with its negative literals ¬v ∈ c. Note that each s(·) stands for some arbitrary but unique

name in its respective context here.

Finally, we represent the set of scenarios (Gi, Ci) for 1 ≤ i ≤ n using predicates

scenario/1, goal/3, and constrained/3. The facts scenario(i) for 1 ≤ i ≤ n de-

note the scenarios to consider. The facts goal(i,v,1) and constrained(i,w,1) for

positive goal literals v ∈ Gi ∩ V and positive constrained literals w ∈ Ci ∩ V and facts

goal(i,v,-1) and constrained(i,w,-1) for negative goal literals ¬v ∈ Gi and negative

constrained literals ¬w ∈ Ci denote the respective intervention goals and side constraints

in each scenario.

Listing 1 shows the instance representation of our toy example logical network in Fig-

ure 1 together with the two intervention scenarios (G1, C1) = ({o1 7→ f , o2 7→ t}, {i1 7→
t}) and (G2, C2) = ({a 7→ t}, ∅).

Listing 1. Toy example problem instance
1 variable(i1). variable(i2). variable(o2). variable(o1). variable(a).
2 variable(b). variable(c). variable(d). variable(e). variable(f).
3 variable(g).
4

5 candidate(i2). candidate(b). candidate(c). candidate(d).
6 candidate(e). candidate(f). candidate(g).
7

8 formula(a,0). formula(b,2). formula(c,1). formula(d,4). formula(e,3).
9 formula(f,6). formula(g,5). formula(o1 ,4). formula(o2 ,7).

10

11 dnf(0,5). dnf(1,6). dnf(1,0). dnf(2,3). dnf(3,7).
12 dnf(4,1). dnf(5,2). dnf(6,4). dnf(6,6). dnf(7,4).
13

14 clause(0,b,1). clause(1,c,1). clause(2,f,1). clause(3,a,1).
15 clause(3,i1 ,1). clause(4,g,1). clause(5,d,-1). clause(6,e,1).
16 clause(7,i2 ,1). clause(7,i1 ,-1).
17

18 scenario (1). scenario (2).
19

20 constrained (1,i1 ,1).
21

22 goal(1,o1 ,-1). goal(1,o2 ,1). goal(2,a,1).

3.2 Logic program

Next we describe our encoding for solving the minimal intervention set problem as de-

scribed in Section 2. Our ASP encoding is shown in Listing 2.

Listing 2. Logic program for solving the minimal intervention set problem
1 goal(T,S) :- goal(_,T,S).
2 goal(T) :- goal(T,_).
3 constrained(Z,E) :- constrained(Z,E,_).
4 constrained(E) :- constrained(_,E).
5

6 satisfy(V,W,S) :- formula(W,D), dnf(D,C), clause(C,V,S).
7 closure(V,T) :- goal(V,T).
8 closure(V,S*T) :- closure(W,T), satisfy(V,W,S), not goal(V,-S*T).
9

10 { intervention(V,S) : closure(V,S) : candidate(V) }.
11 :- intervention(V,1), intervention(V,-1).
12 intervention(V) :- intervention(V,S).
13

14 eval(Z,V,S) :- scenario(Z), intervention(V,S).
15 eval(Z,E,S) :- constrained(Z,E,S), not intervention(E).
16 free(Z,V,D) :- formula(V,D), scenario(Z),
17 not constrained(Z,V), not intervention(V).
18

19 eval_clause(Z,C,-1) :- clause(C,V,S), eval(Z,V,-S).
20

21 eval(Z,V, 1) :- free(Z,V,D), eval(Z,W,T) : clause(C,W,T), dnf(D,C).
22 eval(Z,V,-1) :- free(Z,V,D), eval_clause(Z,C,-1) : dnf(D,C).
23

24 :- goal(Z,T,S), not eval(Z,T,S).
25

26 #const max=0.
27 :- max >0, max + 1 #count{ intervention(_) }.
28

29 #minimize { intervention(_) }.

Note that in the following we use 1 and −1 for truth assignments to t and f , respec-

tively. Furthermore, undefined variables are represented by the absence of assignments to

both t and f . In Lines 1-4 we define auxiliary domain predicates used in the remainder

of the encoding.

Lines 6-8 deserve closer attention since they allow us to reduce significantly the search

space of candidate solutions. We incorporate a preprocessing step introduced in (Sam-

aga et al. 2010) that prunes variable assignments that can never be part of a minimal

intervention set. The idea is to inductively collect all assignments that could be used

to support a goal. First we gather all assignments that make a literal in a clause true

and associate it with variable of the associated DNF (Line 6). Starting from the assign-

ments that can satisfy a goal literal directly (Line 7), we inductively consider variable

assignments (Line 8) that can support the assignments collected so far.

Let us illustrate this on our toy example. In order to satisfy goal(1,o2,1), one would

never consider to intervene variables f or g negatively. Since both reach o2 positively, only

positive interventions on them could help. The same happens for variable e. However,

since e also reaches o1 positively and we have goal(1,o1,-1), a negative intervention of

e could help for this goal.

Next, we use a choice rule in Line 10 to generate candidate solutions. We only choose

interventions collected in the preprocessing step above. The integrity constraint in Line 11

eliminates contradictory interventions. Whereas Line 12 projects the intervention set to

the intervened variables regardless of their signature. For example, one could generate

the intervention set consisting of intervention(e,1) and intervention(c,-1).

Next, we describe in lines 14-15 which variables are clamped in the network according

to the side constraints C in each scenario and the interventions I, i.e., (V, φ|C◦I). Fol-

lowing the previous example, this will generate predicates eval(1,i1,1), eval(1,e,1),

eval(2,e,1), eval(1,c,-1) and eval(2,c,-1). The rule in Line 16 captures the re-

maining variables that have not been clamped i.e., predicates free/3. Then in Line 19 we

declare the fact that a conjunction evaluates to f if at least one of the literals occurring in

it evaluates to f . Afterwards, the rules in Lines 21 and 22 inductively propagate the truth

values t and f in the standard way starting from the clamped truth values. Exploiting

inductive definitions is ASP, this allows us to compute for each scenario the fixpoint of

Ω(V,φ|C◦I). For our example, we can see how the positive intervention over e is propa-

gated as follows. Since variable f is not intervened, its formula (f 7→ e∨ g) described by

predicates formula(f,6), dnf(6,4), dnf(6,6), clause(4,g,1) and clause(6,e,1), is

“free” in all scenarios, i.e., free(1,f,6) and free(2,f,6). Further, given that we have

dnf(6,6) related only to clause(6,e,1) and e was intervened positively, the truth value

for f is propagated in all scenarios regardless of g, i.e., eval(1,f,1) and eval(2,f,1).

Analogously, truth values are propagated through all the network until predicates eval/3

describe the fixpoint in every scenario.

Line 24 declares an integrity constraint to eliminate solutions that do not satisfy all

goals in each scenario. The statements in Line 26 and 27 allows us to optionally bound

the problem by considering only intervention sets up to a given size. And finally, the

minimize constraint in Line 29 denotes that we are interested in solutions involving a

minimal number of interventions.

3.3 Solving

Using a standard ASP solver such as clasp (Gebser et al. 2007), our encoding will actu-

ally generate cardinality minimal models whereas we are interested in inclusion minimal

models. Towards this end, we evaluate two alternative solvers derived from clasp, namely

claspD (Gebser et al. 2013) and hclasp (Gebser et al. 2013). The disjunctive solver claspD

together with the encodings of the metasp framework (Gebser et al. 2011) can easily be

used to compute inclusion minimal models.5 On the other hand, hclasp is a recently de-

veloped solver incorporating domain-specific heuristics into the input language allowing

us to compute inclusion minimal models too. Thus, in order to use hclasp, the minimize

constraint in the last line of the encoding in Listing 2 must be replaced by the rule in

Listing 3.

Listing 3. Changes to Listing 2 in order to use hclasp
29 _heuristic(intervention(V),false ,1) :- closure(V,S), candidate(V).

Together with hclasp’s option --heuristic=domain, the effect of the rule in Listing 3 is

to tell the solver that atoms of the form intervention(V) must be chosen first and as-

signed truth value f . This guarantees that the first answer set found is inclusion minimal

5 The encodings are available at http://www.cs.uni-potsdam.de/wv/metasp/

regarding the set of atoms in focus (Castell et al. 1996). To further ensure inclusion mini-

mality for all subsequent answer sets, a constraint must be added excluding the previous

answer set (cf. (Di Rosa et al. 2010)); this is invoked by hclasp’s option --enum-mode

record.

In Listing 4 we show the models found for the toy instance described in Listing 1 using

claspD. Of course, the same models are found with hclasp using the following command

line: gringo enc-hclasp.lp toy.lp | hclasp --heuristic=domain --enum-mode record 0

Listing 4. Computing all MISs for the toy instance
$ gringo --reify encoding.lp toy.lp |\

gringo - meta.lp metaD.lp metaO.lp <(echo "optimize (1,1,incl).") |\
claspD 0

claspD version 2
Reading from stdin
Solving ...
Answer: 1
intervention(i2 ,-1) intervention(b,-1) intervention(f,1)
Answer: 2
intervention(e,-1) intervention(b,-1) intervention(f,1)
Answer: 3
intervention(e,-1) intervention(b,-1) intervention(g,1)
Answer: 4
intervention(i2 ,-1) intervention(b,-1) intervention(g,1)
Answer: 5
intervention(b,-1) intervention(f,1) intervention(d,-1)
Answer: 6
intervention(b,-1) intervention(d,-1) intervention(g,1)
Answer: 7
intervention(g,1) intervention(c,-1)
Answer: 8
intervention(e,1) intervention(c,-1)
Answer: 9
intervention(f,1) intervention(c,-1)
SATISFIABLE

Models : 9
Time : 0.034s (Solving: 0.00s 1st Model: 0.00s Unsat: 0.00s)
CPU Time : 0.010s

4 Benchmarks

We evaluate the performance and scalability of our ASP solution over four real-world

and biologically relevant benchmarks (Table 1). Three of these benchmarks (EGFR,

EGFR multiple and TCR) were used in (Samaga et al. 2010) and their corresponding

logical networks were recently published (Saez-Rodriguez et al. 2007; Samaga et al. 2009).

Further, we also use a larger unpublished logical network (TBH6b) provided by Axel von

Kamp and Steffen Klamt. While the authors in (Samaga et al. 2010) restricted their

study to a maximum cardinality of 3, herein we extend this limitation to a maximum

cardinality of 10 or no limit at all. For each case we report the number of ⊆-minimal

intervention sets (MISs) and CPU time in seconds for each solver. Computations were run

on a MacBook Pro, Intel Core i7, 2.7 GHz and 4 GB of RAM using gringo-3.0.3 together

with claspD-2 and hclasp, respectively. In Table 1 we describe our benchmark problem

instances. For each case we report the number of variables (i.e. species) in the logical

network, the number of intervention scenarios, the number of intervention constraints,

Table 1. Benchmark instances of the multi-scenario intervention set problem
Instance Variables Scenarios Side Constraints Goals Candidate MISs

(Gj , Cj)j∈J wrt (V, φ) |V | |J |
∑

j∈J |Cj |
∑

j∈J |Gj |
EGFR 103 1 28 2 313 × 219

EGFR multiple 103 34 974 408 352 × 215

TCR 94 1 17 12 335 × 228

TBH6b 203 1 17 3 385 × 224

Table 2. MISs calculation up to a given size (k) or unbounded (∞) with a timeout set to

1000 sec. For each instance we report the number of intervention sets (I) and CPU time

(sec) of both, claspD and hclasp.

EGFR EGFR multiple TCR TBH6b
k I claspD hclasp I claspD hclasp I claspD hclasp I claspD hclasp
1 15 0.02 0.00 0 0.86 0.07 0 0.04 0.00 0 0.75 0.00
2 21 0.11 0.00 0 0.93 0.08 0 0.05 0.00 0 0.81 0.00
3 21 0.13 0.00 8 1.15 0.10 7 0.05 0.00 6 0.92 0.00
4 21 0.14 0.00 38 11.16 0.15 26 0.36 0.03 6 18.00 0.00
5 21 0.17 0.00 74 41.57 0.21 1196 2.04 0.08 6 17.40 0.00
6 21 0.16 0.00 83 73.14 0.19 4290 60.38 0.37 15 20.28 0.00
7 21 0.16 0.00 83 90.66 0.20 7258 198.19 0.35 15 47.19 0.00
8 21 0.16 0.00 83 90.56 0.24 8776 465.67 0.75 24 52.11 0.01
9 21 0.18 0.00 83 88.95 0.25 9316 655.22 1.09 207 90.67 0.04

10 21 0.19 0.00 83 84.60 0.23 9704 669.64 0.94 1248 667.85 0.23
∞ 21 0.13 0.00 83 91.83 0.19 13016 512.52 1.19 -a - -

a Both solvers have reached a timeout for this case, however their behaviors are very different. After
1000 seconds, claspD found only 648 models whereas hclasp found 894483 models

the number of intervention goals, and the number of candidate intervention sets. The

number of candidate intervention sets is computed as follows. Let n be the number of

variables that could be intervened positively and negatively. Let m be the number of

variables that could be intervened either positively or negatively, but not both. Then,

the number of intervention sets is given by 3n × 2m. We note that for EGFR, TCR

and TBH6b interventions are forbidden over constrained and goal variables, whereas for

EGFR multiple this is not the case (there are no forbidden variables). This relaxation is

necessary in order to reproduce the results reported in (Samaga et al. 2010) and compare

both methods. Nonetheless, it worth noting that there are no solutions satisfying the 34

scenarios. In fact, depending on the application at hand, finding no solution could be

more interesting than relaxing the problem. In Table 2 we show the performance over

the four problem instances. First, we were able to compute all MISs of up to size 10

for all instances regardless of the solver used. This represents a significant improvement

compared to previous approaches limited in practice to compute intervention sets having

maximum cardinality of 3 or 4. Moreover, for EGFR, EGFR multiple and TCR we were

able to solve the unbounded problem, i.e. k = ∞. That is, for the mentioned instances,

we are able to completely characterize all feasible inclusion minimal intervention sets.

Further, while for k ≤ 4 the difference between claspD and hclasp may not be very

evident, for 5 ≤ k ≤ 10 computation times for claspD tend to grow significantly whereas

computation times for hclasp remain relatively constant. Finally, for the unbounded case

of TBH6b where we found a timeout after 1000 seconds, it is interesting to note the

different behavior of each solver (see footnote in Table 2). One reason for the worse

performance of claspD is related to the way metasp implements subset minimization.

Currently, for non-tight programs, the encoding of inclusion minimality in metasp leads

to a quadratic blowup in the propositional program passed to claspD. Compare with

the EGFR instance where the underlying network is acyclic and hence the resulting

propositional program is tight. The performance of claspD is better on this instance.

Overall, hclasp’s solution recording based algorithm appears to be the better choice for

enumerating subset minimal answers.

5 Discussion and Conclusion

Logic modeling is an emerging qualitative approach to capture mechanistic behavior in

large-scale biological systems. Despite of its relative simplicity, it allows for addressing

relevant problems related to drug target identification, experimental design, and diagno-

sis. In this context, finding minimal intervention strategies in a logic signaling network

with desired outcomes leads to challenging combinatorial problems that require advanced

solving technologies. Previous work on this subject consists of dedicated algorithms and

special purpose search space reduction techniques for coping with combinatorial explo-

sions. In fact, in practice, such algorithms are limited to searching intervention sets having

only a small number of interventions (eg. ≤ 3).

In this work, we have provided a precise characterization of the minimal intervention

set problem relying on Kleene’s three-valued logic and fixpoint semantics (close to tradi-

tional logic programming concepts). In this context, our fixpoint characterization allows

us to capture the steady states of a logical network following from a set of clamped

values. We have proposed an ASP encoding for this problem and we have evaluated its

performance using real-world biological benchmarks. Negation by default and the recur-

sive definition of reachability make of ASP a very suitable framework for this problem.

For addressing our problem’s complexity, we have used and compared the ASP solvers

claspD and hclasp.

Our ASP encoding incorporates a search space reduction based on the interaction

graph underlying a logical network (Samaga et al. 2010). This can be exploited during

the grounding phase, significantly reducing the number of candidate solutions. In fact,

we considered also other special purpose techniques from the aforecited work but did

not observe any improvements in performance on the available instances. Experiments

have shown that our approach outperforms the previous dedicated algorithms in up to

four orders of magnitude (for small number of interventions (≤ 3) still feasible for the

algorithm). This was not very surprising since such algorithms are based on a standard

breadth-first search using additional techniques for search space reduction. More impor-

tantly, we are able to search for significantly larger intervention sets or even solve the

unbounded problem (ie. no limit in the number of interventions). While considering a

small number of interventions the number of solutions (i.e. intervention sets) is in the

order of tens, with a larger number of interventions we have found thousands of fea-

sible solutions. Furthermore, being able to solve the unbounded problem allows us to

completely characterize the set of solutions.

Nowadays, in signaling networks, large-scale (> 10) interventions combining different

inhibitors can be considered as a long-term technological perspective. Nevertheless, in

2 out of 4 benchmarks (EGFR and EGFR mutiple), we did not find intervention sets

of size bigger than 10. This suggests that extending the set of interventions may not

be interesting for such models. Similarly, for the remaining benchmark where we have

solved the unbounded problem (TCR), ∼ 70% of the minimal intervention sets have size

smaller than 10. This provides also a useful information on the flexibility of the system.

On the contrary, knowing that a large number of interventions are required to reach

certain state could help to understand (at least theoretically) the systems’ robustness.

Altogether, being able to compute both small and large admissible intervention sets,

appears as an interesting and relevant feature of our approach.

Both computational and biological perspective tracks are open. On the computational

side, a precise estimation of the empirical complexity appears to be non trivial and very

specific to each instance. Given a problem instance, the number of candidate solutions

can be computed analytically. However, experiments have shown that this is not the only

parameter determining the empirical complexity. The number of goal variables and their

location in the logical network, the number of intervention scenarios, and topological

properties of the network may have an impact on the computational efforts required in

practice. More generally, it will be interesting to investigate in how far the employed

concepts from logic programming furnish an adequate and general tool for addressing

problems in (Boolean) networks, as also indicated by the work of Inoue and Sakama

in (Inoue 2011; Inoue and Sakama 2012). On the biological side, in the light of such a

large number of solutions, the way to select among them arises. In general, when the

inherent noise is taken into account, several logical networks can describe a biological

system equally (or similarly) well (Saez-Rodriguez et al. 2009). Thus, one could extend

the intervention set problem to a family of plausible logical networks. This way, we would

reduce the number of solutions by selecting the more robust of them. That is, intervention

sets satisfying each scenario in all networks.

